

NAG Library Routine Document

G03FCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G03FCF performs non-metric (ordinal) multidimensional scaling.

2 Specification

```

SUBROUTINE G03FCF (TYP, N, NDIM, D, X, LDX, STRESS, DFIT, ITER, IOPT,      &
                  WK, IWK, IFAIL)
INTEGER           N, NDIM, LDX, ITER, IOPT, IWK(N*(N-1)/2+N*NDIM+5),      &
                  IFAIL
REAL (KIND=nag_wp) D(N*(N-1)/2), X(LDX,NDIM), STRESS, DFIT(2*N*(N-1)),    &
                  WK(15*N*NDIM)
CHARACTER(1)     TYP

```

3 Description

For a set of n objects, a distance or dissimilarity matrix D can be calculated such that d_{ij} is a measure of how ‘far apart’ the objects i and j are. If p variables x_k have been recorded for each observation this measure may be based on Euclidean distance, $d_{ij} = \sum_{k=1}^p (x_{ki} - x_{kj})^2$, or some other calculation such as the number of variables for which $x_{kj} \neq x_{ki}$. Alternatively, the distances may be the result of a subjective assessment. For a given distance matrix, multidimensional scaling produces a configuration of n points in a chosen number of dimensions, m , such that the distance between the points in some way best matches the distance matrix. For some distance measures, such as Euclidean distance, the size of distance is meaningful, for other measures of distance all that can be said is that one distance is greater or smaller than another. For the former metric scaling can be used, see G03FAF, for the latter, a non-metric scaling is more appropriate.

For non-metric multidimensional scaling, the criterion used to measure the closeness of the fitted distance matrix to the observed distance matrix is known as STRESS. STRESS is given by,

$$\sqrt{\frac{\sum_{i=1}^n \sum_{j=1}^{i-1} (\hat{d}_{ij} - \tilde{d}_{ij})^2}{\sum_{i=1}^n \sum_{j=1}^{i-1} \hat{d}_{ij}^2}}$$

where \hat{d}_{ij}^2 is the Euclidean squared distance between points i and j and \tilde{d}_{ij} is the fitted distance obtained when \hat{d}_{ij} is monotonically regressed on d_{ij} , that is \tilde{d}_{ij} is monotonic relative to d_{ij} and is obtained from \hat{d}_{ij} with the smallest number of changes. So STRESS is a measure of by how much the set of points preserve the order of the distances in the original distance matrix. Non-metric multidimensional scaling seeks to find the set of points that minimize the STRESS.

An alternate measure is squared STRESS, *sstress*,

$$\sqrt{\frac{\sum_{i=1}^n \sum_{j=1}^{i-1} (\hat{d}_{ij}^2 - \tilde{d}_{ij}^2)^2}{\sum_{i=1}^n \sum_{j=1}^{i-1} \hat{d}_{ij}^4}}$$

in which the distances in STRESS are replaced by squared distances.

In order to perform a non-metric scaling, an initial configuration of points is required. This can be obtained from principal coordinate analysis, see G03FAF. Given an initial configuration, G03FCF uses the optimization routine E04DGF/E04DGA to find the configuration of points that minimizes STRESS or *sstress*. The routine E04DGF/E04DGA uses a conjugate gradient algorithm. G03FCF will find an optimum that may only be a local optimum, to be more sure of finding a global optimum several different initial configurations should be used; these can be obtained by randomly perturbing the original initial configuration using routines from Chapter G05.

4 References

Chatfield C and Collins A J (1980) *Introduction to Multivariate Analysis* Chapman and Hall

Krzanowski W J (1990) *Principles of Multivariate Analysis* Oxford University Press

5 Arguments

- 1: TYP – CHARACTER(1) *Input*
On entry: indicates whether STRESS or *sstress* is to be used as the criterion.
TYP = 'T'
STRESS is used.
TYP = 'S'
sstress is used.
Constraint: TYP = 'S' or 'T'.
- 2: N – INTEGER *Input*
On entry: *n*, the number of objects in the distance matrix.
Constraint: N > NDIM.
- 3: NDIM – INTEGER *Input*
On entry: *m*, the number of dimensions used to represent the data.
Constraint: NDIM ≥ 1.
- 4: D(N × (N – 1)/2) – REAL (KIND=nag_wp) array *Input*
On entry: the lower triangle of the distance matrix *D* stored packed by rows. That is D((*i* – 1) × (*i* – 2)/2 + *j*) must contain *d*_{*ij*}, for *i* = 2, 3, ..., *n* and *j* = 1, 2, ..., *i* – 1. If *d*_{*ij*} is missing then set *d*_{*ij*} < 0; for further comments on missing values see Section 9.
- 5: X(LDX, NDIM) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the *i*th row must contain an initial estimate of the coordinates for the *i*th point, for *i* = 1, 2, ..., *n*. One method of computing these is to use G03FAF.
On exit: the *i*th row contains *m* coordinates for the *i*th point, for *i* = 1, 2, ..., *n*.

- 6: LDX – INTEGER *Input*
On entry: the first dimension of the array X as declared in the (sub)program from which G03FCF is called.
Constraint: $LDX \geq N$.
- 7: STRESS – REAL (KIND=nag_wp) *Output*
On exit: the value of STRESS or *sstress* at the final iteration.
- 8: DFIT($2 \times N \times (N - 1)$) – REAL (KIND=nag_wp) array *Output*
On exit: auxiliary outputs.
 If TYP = 'T', the first $n(n - 1)/2$ elements contain the distances, \hat{d}_{ij} , for the points returned in X, the second set of $n(n - 1)/2$ contains the distances \hat{d}_{ij} ordered by the input distances, d_{ij} , the third set of $n(n - 1)/2$ elements contains the monotonic distances, \tilde{d}_{ij} , ordered by the input distances, d_{ij} and the final set of $n(n - 1)/2$ elements contains fitted monotonic distances, \tilde{d}_{ij} , for the points in X. The \tilde{d}_{ij} corresponding to distances which are input as missing are set to zero.
 If TYP = 'S', the results are as above except that the squared distances are returned.
 Each distance matrix is stored in lower triangular packed form in the same way as the input matrix *D*.
- 9: ITER – INTEGER *Input*
On entry: the maximum number of iterations in the optimization process.
 ITER = 0
 A default value of 50 is used.
 ITER < 0
 A default value of $\max(50, 5nm)$ (the default for E04DGF/E04DGA) is used.
- 10: IOPT – INTEGER *Input*
On entry: selects the options, other than the number of iterations, that control the optimization.
 IOPT = 0
 Default values are selected as described in Section 9. In particular if an accuracy requirement of $\epsilon = 0.00001$ is selected, see Section 7.
 IOPT > 0
 The default values are used except that the accuracy is given by 10^{-i} where $i = IOPT$.
 IOPT < 0
 The option setting mechanism of E04DGF/E04DGA can be used to set all options except **Iteration Limit**; this option is only recommended if you are an experienced user of NAG optimization routines. For further details see E04DGF/E04DGA.
- 11: WK($15 \times N \times NDIM$) – REAL (KIND=nag_wp) array *Workspace*
- 12: IWK($N \times (N - 1)/2 + N \times NDIM + 5$) – INTEGER array *Workspace*
- 13: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the

recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NDIM < 1,
or $N \leq \text{NDIM}$,
or TYP \neq 'T' or 'S',
or LDX < N.

IFAIL = 2

On entry, all elements of $D \leq 0.0$.

IFAIL = 3

The optimization has failed to converge in ITER function iterations. Try either increasing the number of iterations using ITER or increasing the value of ϵ , given by IOPT, used to determine convergence. Alternatively try a different starting configuration.

IFAIL = 4

The conditions for an acceptable solution have not been met but a lower point could not be found. Try using a larger value of ϵ , given by IOPT.

IFAIL = 5

The optimization cannot begin from the initial configuration. Try a different set of points.

IFAIL = 6

The optimization has failed. This error is only likely if IOPT < 0. It corresponds to IFAIL = 4, 7 and 9 in E04DGF/E04DGA.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

After a successful optimization the relative accuracy of STRESS should be approximately ϵ , as specified by IOPT.

8 Parallelism and Performance

G03FCF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The optimization routine E04DGF/E04DGA used by G03FCF has a number of options to control the process. The options for the maximum number of iterations (**Iteration Limit**) and accuracy (**Optimality Tolerance**) can be controlled by ITER and IOPT respectively. The printing option (**Print Level**) is set to -1 to give no printing. The other option set is to stop the checking of derivatives (**Verify = NO**) for efficiency. All other options are left at their default values. If however $\text{IOPT} < 0$ is used, only the maximum number of iterations is set. All other options can be controlled by the option setting mechanism of E04DGF/E04DGA with the defaults as given by that routine.

Missing values in the input distance matrix can be specified by a negative value and providing there are not more than about two thirds of the values missing the algorithm may still work. However the routine G03FAF does not allow for missing values so an alternative method of obtaining an initial set of coordinates is required. It may be possible to estimate the missing values with some form of average and then use G03FAF to give an initial set of coordinates.

10 Example

The data, given by Krzanowski (1990), are dissimilarities between water vole populations in Europe. Initial estimates are provided by the first two principal coordinates computed.

10.1 Program Text

```

Program g03fcfe
!      G03FCF Example Program Text
!
!      Mark 26 Release. NAG Copyright 2016.
!
!      .. Use Statements ..
Use nag_library, Only: g03faf, g03fcf, nag_wp, x04caf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: stress
Integer                    :: ifail, iopt, iter, ld, ldfit, ldx,    &
                          liwk, lwk, n, ndim
Character (1)              :: roots, typ
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: d(:), dfit(:), eval(:), wk(:),    &
                          x(:, :)
Integer, Allocatable       :: iwkw(:)
!      .. Intrinsic Procedures ..
Intrinsic                  :: max
!      .. Executable Statements ..
Write (nout,*) 'G03FCF Example Program Results'
Write (nout,*)

```

```

!      Skip heading in data file
      Read (nin,*)

!      Read in the problem size
      Read (nin,*) n, ndim, roots, typ

      ld = n*(n-1)/2
      ldx = n
      lwk = max(n*(n+17)/2-1,15*n*ndim)
      liwk = max(5*n,n*(n-1)/2+n*ndim+5)
      ldfit = 2*n*(n-1)
      Allocate (d(ld),x(ldx,ndim),eval(n),wk(lwk),iwk(liwk),dfit(ldfit))

!      Read in the lower triangular part of the distance matrix
      Read (nin,*) d(1:ld)

!      Perform principal co-ordinate analysis
      ifail = 0
      Call g03faf(roots,n,d,ndim,x,ldx,eval,wk,iwk,ifail)

!      Use default values for number of iterations and options
      iter = 0
      iopt = 0

!      Perform multi-dimensional scaling
      ifail = 0
      Call g03fcf(typ,n,ndim,d,x,ldx,stress,dfit,iter,iopt,wk,iwk,ifail)

!      Display the results
      Write (nout,99999) 'STRESS = ', stress
      Write (nout,*)
      Flush (nout)
      ifail = 0
      Call x04caf('General',' ',n,ndim,x,ldx,'Co-ordinates',ifail)

99999 Format (10X,A,E13.4)
      End Program g03fcfe

```

10.2 Program Data

G03FCF Example Program Data

```

14 2 'L' 'T'                                     :N,NDIM,ROOTS,TYP
0.099
0.033 0.022
0.183 0.114 0.042
0.148 0.224 0.059 0.068
0.198 0.039 0.053 0.085 0.051
0.462 0.266 0.322 0.435 0.268 0.025
0.628 0.442 0.444 0.406 0.240 0.129 0.014
0.113 0.070 0.046 0.047 0.034 0.002 0.106 0.129
0.173 0.119 0.162 0.331 0.177 0.039 0.089 0.237 0.071
0.434 0.419 0.339 0.505 0.469 0.390 0.315 0.349 0.151 0.430
0.762 0.633 0.781 0.700 0.758 0.625 0.469 0.618 0.440 0.538 0.607
0.530 0.389 0.482 0.579 0.597 0.498 0.374 0.562 0.247 0.383 0.387 0.084
0.586 0.435 0.550 0.530 0.552 0.509 0.369 0.471 0.234 0.346 0.456 0.090 0.038 :D

```

10.3 Program Results

G03FCF Example Program Results

STRESS = 0.1256E+00

Co-ordinates

	1	2
1	0.2060	0.2438
2	0.1063	0.1418
3	0.2224	0.0817
4	0.3032	0.0355
5	0.2645	-0.0698

6	0.1554	-0.0435
7	-0.0070	-0.1612
8	0.0749	-0.3275
9	0.0488	0.0289
10	0.0124	-0.0267
11	-0.1649	-0.2500
12	-0.5073	0.1267
13	-0.3093	0.1590
14	-0.3498	0.0700

